
	WWARN is committed to supporting efficient and quality data collection and analysis for antimalarial drug research. As such we are sharing the NONMEM code used for this publication. NONMEM is a computer program, NONlinear Mixed Effects Modeling, used by WWARN’s Pharmacometric Scientific Group. 
For further information, please contact info@wwarn.org 

	Please use the following reference:  
WWARN NONMEM file used to develop:  Population Pharmacokinetics of the Antimalarial Amodiaquine: a Pooled Analysis To Optimize Dosing. 
Ali AM, Penny MA, Smith TA, Workman L, Sasi P, Adjei GO, Aweeka F, Kiechel JR, Jullien V, Rijken MJ, McGready R, Mwesigwa J, Kristensen K, Stepniewska K, Tarning J, Barnes KI, Denti P; WWARN Amodiaquine PK Study Group.
Antimicrob Agents Chemother. 2018;62(10). pii: e02193-17.

	
$INPUT     
    ID 	; Patient ID
    TIME 	; Time of sample 
    DV 	; Dependent variable (natural logarithm of observed concentrations, nmol/L)
    WT 	; Body weight (covariate, kg) 
    AGE	; Age (covaraite, years)
    EVID 	; Event ID record 
    MDV 	; Missing dependent variable (1=missing)
    AMT 	; Dose amount (μmol)
    CMT 	; Compartment (1=dose, 2=amodiaquine (AQ), 4=desethylamodiaquine (DAQ)) 
    OCC	; Occasion (1=first dose, 2=second dose , 3=third dose) 
    LOQ	; Limit of quantification

$DATA  
    dataset.csv  IGNORE=#

$SUBROUTINE  
    ADVAN13  TOL=8  TRANS1

$MODEL      
    COMP = (1) 	 ; Dose
    COMP = (2) 	 ; AQ central compartment 
    COMP = (3) 	 ; AQ peripheral compartment 
    COMP = (4) 	 ; DAQ  central compartment  
    COMP = (5) 	 ; DAQ first peripheral compartment 
    COMP = (6) 	 ; DAQ second peripheral compartment

$PRIOR      
    NWPRI NPEXP = 1 PLEV = 0.9999	; Prior information for the maturation function
    $THETAP  0.02 FIX			; Log postmenstrual age to reach 50% of full maturation for AQ
    $THETAP  1.25 FIX			; Log shape parameter for AQ
    $THETAP  0.02 FIX 			; Log postmenstrual to reach 50% of full maturation for DAQ
    $THETAP  1.25 FIX 			; Log shape parameter for DAQ
    $THETAPV  BLOCK(4) FIX		; THETA BLOCK
           0.01
           0 0.01
           0 0 0.01
           0  0 0 0.01

$PK
;------------------------------------Age covariate  for AQ -------------------------------------------------------------------------
    PGA = AGE+(9/12)	; Postmenstrual age (years)
    MEDAGE = 8.1	; Median age (years)  
    TV_PGA = MEDAGE+(9/12)	; Median postmenstrual age (years) 
    LOGPGA50 = THETA(1)	; Log postmenstrual age to reach 50% of full maturation for AQ
    GAMMA1 = THETA(2)	; Shape parameter for AQ
    MATCL = 1/(1+EXP(-GAMMA1*(LOG(PGA)-LOGPGA50)))* (1+EXP(-GAMMA1*(LOG(TV_PGA)-
        LOGPGA50)))	; Postmenstrual age-maturation relationship for AQ	
;----------------------------------------------------------------------------------------------------------------------------------------


;------------------------------------Age covariate  for DAQ ----------------------------------------------------------------------
    LOGPGA50_DAQ = THETA(3) 		; Log postmenstrual age to reach 50% of full maturation for DAQ
    GAMMA2 = THETA(4) 			; Shape parameter for DAQ
    MATCLDAQ = 1/(1+EXP(-GAMMA2*(LOG(PGA)-LOGPGA50_DAQ)))*(1+EXP(-GAMMA2* 
        (LOG(TV_PGA)-LOGPGA50_DAQ))) 	; Postmenstrual age-maturation relationship for DAQ
;----------------------------------------------------------------------------------------------------------------------------------------

;------------------------------------ First dose covariate --------------------------------------------------------------------------
    IF (OCC.GT.1)    OCC_BIO = 1		; Dose effect 
    IF (OCC.EQ.1)    OCC_BIO = 1 + THETA(19)	; Proportional effect for first dose
;----------------------------------------------------------------------------------------------------------------------------------------

;------------------------------------ Between occasion variability (BOV) -----------------------------------------------------
    BOV_MTT = ETA(11)	; BOV for mean transit time
    IF (OCC.EQ.2) BOV_MTT = ETA(12) 	; Occasion 2
    IF (OCC.EQ.3) BOV_MTT = ETA(13)	; Occasion 3
    BOV_KA = ETA(14) 	; BOV for absorption rate
    IF(OCC.EQ.2) BOV_KA = ETA(15)	; Occasion 2 
    IF(OCC.EQ.3) BOV_KA = ETA(16)	; Occasion 3 
    BOV_BIO = ETA(17)  	; BOV for relative bioavailability
    IF (OCC.EQ.2) BOV_BIO = ETA(18) 	; Occasion 2  
    IF (OCC.EQ.3) BOV_BIO = ETA(19) 	; Occasion 3  
;----------------------------------------------------------------------------------------------------------------------------------------
    TVCL = THETA(5) * ((WT/49)**0.75) * MATCL	; Population AQ clearance
    CL = TVCL * EXP(ETA(1)) 	; Individual AQ clearance

    TVV2 = THETA(6) * (WT/49)	; Population AQ central volume
    V2 = TVV2 * EXP(ETA(2))	; Individual AQ central volume

    TVQ1 = THETA(7) * ((WT/49)**0.75)            	; Population AQ inter-compartment clearance
    Q1 = TVQ1 * EXP(ETA(3))	; Individual AQ inter-compartment clearance

    TVV3 = THETA(8) * (WT/49)	; Population AQ peripheral volume
    V3 = TVV3 * EXP(ETA(4))	; Individual AQ peripheral volume

    TVKA = EXP(THETA(9))  	; Population absorption rate constant
    KA = TVKA * EXP(BOV_KA)	; Individual absorption rate constant

    TVMTT = THETA(10)  	; Population mean transit time
    MTT = TVMTT * EXP(BOV_MTT)	; Individual mean transit time

    TVNN = EXP(THETA(11))  	; Population number of transit compartments
    NN = TVNN	; Individual number of transit compartments

    TVBIO = THETA(12) * OCC_BIO	; Population relative bioavailability
    BIO = TVBIO * EXP(BOV_BIO)	; Individual relative bioavailability

    TVV4 = THETA(13) * (WT/49)	; Population DAQ central volume
    V4 = TVV4 * EXP(ETA(5))	; Individual DAQ central volume

    TVCLM = THETA(14)*((WT/49)**0.75)*MATCLDAQ	; Population DAQ clearance
    CLM = TVCLM * EXP(ETA(6))	; Individual DAQ clearance

    TVQ2 = THETA(15) * ((WT/49)**0.75)	; Population DAQ inter-compartment clearance1
    Q2 = TVQ2* EXP(ETA(7))	; Individual DAQ inter-compartment clearance1

    TVV5 = THETA(16) * (WT/49)	; Population DAQ peripheral volume 1
    V5 = TVV5 * EXP(ETA(8))	; Individual DAQ peripheral volume 1

    TVQ3 = THETA(17) * ((WT/49)**0.75)	; Population DAQ inter-compartment clearance2
    Q3 = TVQ2 * EXP(ETA(9))	; Individual DAQ inter-compartment clearance2

    TVV6 = THETA(18) * (WT/49)	; Population DAQ peripheral volume 2
    V6 = TVV6 * EXP(ETA(10))	; Individual DAQ peripheral volume 2

    K23 = Q1/V2	; AQ distribution rate constant (COMP 2 --> 3)
    K32 = Q1/V3	; AQ distribution rate constant (COMP 3 --> 2)
    K24 = CL / V2	; AQ elimination rate constant (COMP 2 --> 4)
    K45 = Q2/V4	; DAQ distribution rate constant (COMP 4 --> 5)
    K54 = Q2/V5	; DAQ distribution rate constant (COMP 5 --> 4)
    K46 = Q3/V4	; DAQ distribution rate constant (COMP 4 --> 6)
    K64 = Q3/V6	; DAQ distribution rate constant (COMP6 --> 4)
    K40 = CLM / V4	; DAQ elimination rate constant (COMP 4 --> 0)
    F1 = 0		; The extent of absorption for  dosing compartment 1
    S2 = V2/1000	; Scaling for AQ central volume
    S4 = V4/1000	; Scaling for DAQ central volume

    IF (NEWIND.NE.2.OR.EVID.GE.3)    THEN		; Beginning of new individual                             
        TNXD = TIME 		; Initial time  
        PNXD = AMT    		; Initial dose
    ENDIF
    TDOS = TNXD      		; Initial time  to variable TDOS
    PD = PNXD			; Initial dose to variable PD

    IF (AMT.GT.0) THEN
        TNXD = TIME		; Time of dose
        PNXD = AMT		; Dose amount	
    ENDIF
    KTR = (NN+1)/MTT	                    	; Transfer rate constant 
    PIZZA = LOG(BIO*PD*KTR + 0.00001) - GAMLN(NN+1)   	; Stirling approximation 

$DES
    TEMPO = T – TDOS			; Time after dose
    INPUT = 0								; Initialization the dose input
    IF (PD.GT.0.AND.TEMPO.GT.0) THEN
        INPUT = EXP(PIZZA+NN*LOG(KTR*TEMPO)-KTR*TEMPO)	; Input of dose
    ENDIF   

    DADT(1) = INPUT - KA*A(1)		; 1 Dose compartment
    DADT(2) = KA*A(1) - K23*A(2) + K32*A(3) -K24*A(2) 		; 2 AQ central compartment
    DADT(3) = K23*A(2) - K32*A(3) 		; 3 AQ peripheral compartment
    DADT(4) = K24* A(2) - K45*A(4) + K54*A(5)- 
        K46*A(4) + K64*A(6) -K40*A(4)			; 4 DAQ central compartment
    DADT(5) = K45*A(4) - K54*A(5)		; 5 DAQ peripheral compartment 1
    DADT(6) = K46*A(4) - K64*A(6)		; 6 DAQ peripheral compartment 2

$ERROR
    IF (CMT = 2) THEN
        IPRED = A(2)/S2           	; Predicted plasma concentration of AQ
        ADD = THETA(20) + LOQ/5	; AQ additive residual error
        PROP = THETA(21)*IPRED	; AQ proportional residual error
    ENDIF

    IF (CMT = 4) THEN
        IPRED = A(4)/S4           	; Predicted plasma concentration of DAQ
        ADD = THETA(22) + LOQ/5	; DAQ additive residual error
        PROP = THETA(23)*IPRED	; DAQ proportional residual error
    ENDIF
    W = SQRT(ADD**2+PROP**2)	; Residual error
    IRES = DV-IPRED	; Individual residue
    IF (W.LE.0.0001) W = 0.0001
    IWRES = IRES/W	; Individual weighted residue
    Y = IPRED + W*EPS(1)  	; Prediction of Y
;----------------------------------------------------------------------------------------------------------------------------------------
$THETA	; Initial estimates of theta
    (-1,-0.017,1.5) 		; 1. Postmenstrual age to reach 50% of full maturation for AQ (log)
    (-1,1.28,2) 			; 2. Shape parameter (log)
    (-1,0.072,1.5) 			; 3 . Postmenstrual age to reach 50% of full maturation for DAQ (log)
    (-1,1.17,2) 			; 4. Shape parameter (log)
    (0,2960,10000) 		; 5. AQ clearance
    (0,13500,30000) 		; 6. AQ central volume of distribution
    (10,2310,10000) 		; 7. AQ inter-compartment clearance
    (0,22700,50000) 		; 8. AQ peripheral volume of distribution
    (-1,-0.529,2) 	 		; 9. Absoprtion rate constant in log scale
    (0.1,0.236,4) 			; 10. Mean transit time
    (0,0.647,5) 			; 11. Number of transit compartments in log scale
    (1) FIX 			; 12. Relative bioavailability
    (10,258,5000) 		; 13. DAQ central volume of distribution
    (10,32.6,1000) 		; 14.  DAQ clearance
    (5,154,1000) 			; 15. DAQ inter-compartment clearance 1 
    (100,2460,5000) 		; 16. DAQ peripheral volume of distribution 1
    (10,31.3,1000) 		; 17. DAQ inter-compartment clearance 2
    (500,5580,10000) 		; 18. DAQ peripheral volume of distribution 2
    (-1,-0.224,6) 			; 19. First dose occasion on bioavailability
    (0.1,0.445,10) 			; 20. Additive residul error for AQ
    (0.1,0.199,0.5) 		; 21. Proportional residue for AQ
    0 FIX 				; 22. Additive residul error for DAQ
    (0.1,0.242,0.7) 		; 23. Proportional residul error for DAQ

$OMEGA			; Initial estimates for omega
    0.104  			; 1. AQ IIV clearance
    0.282				; 2. AQ IIV central volume
    0 FIX				; 3. AQ IIV inter-compartment clearance
    0 FIX 				; 4. AQ IIV peripheral volume
    0.452  			; 5. DAQ IIV central volume
    0.040				; 6. DAQ IIV clearance
    0 FIX 				; 7. DAQ IIV inter-compartment clearance 1
    0 FIX				; 8. DAQ IIV peripheral volume 1
    0 FIX 				; 9. DAQ IIV inter-compartment clearance 2
    0 FIX				; 10. DAQ IIV peripheral volume 2
$OMEGA  BLOCK(1)  0.872  	; 11. BOV for mean transit time
$OMEGA  BLOCK(1) SAME	; 12. Occasion 2
$OMEGA  BLOCK(1) SAME	; 13. Occasion 3
$OMEGA  BLOCK(1)  0.617  	; 14. BOV for absorption rate constant
$OMEGA  BLOCK(1) SAME	; 15. Occasion 2
$OMEGA  BLOCK(1) SAME	; 16. Occasion 3
$OMEGA  BLOCK(1)  0.096	; 17. BOV for bioavailability
$OMEGA  BLOCK(1) SAME	; 18. Occasion 2
$OMEGA  BLOCK(1) SAME	; 19. Occasion 3

$SIGMA	; Initial estimates of sigma
    1  FIX				; Residual variability

$ESTIMATION   POSTHOC MAXEVAL=9999   METHOD=1 INTER

